Pair Distribution Function Refinements using TOPAS v6

Phil Chater
Senior Beamline Scientist, XPDF (I15-1)

diamond

June 2017

Context of presentation

- This presentation describes user macros used to model PDF data using TOPAS v6
- The macros referred to throughout this document are available in pdf.inc
- A jEdit XInsert menu system called pdf.insert.xml is available to creating PDF .inp files

Overview | Features of a PDF

- Peak positions: Bond lengths, geometry, unit cell
- Peak widths: Vibrations, phonons, disorder
- Peak areas: Coordination number, occupancies
- Peak dampening: Size, shape



All of these can be affected by the instrument resolution function (and PDF processing)

Instrumental Functions

Resolution | Reciprocal- and real-space

- An ideal measurement would have no contribution from the instrument resolution
- In practice, PDF measurements are typically done on instruments with medium- to poor-reciprocal-space (Bragg) resolution
  - A wide Q range and high flux is more important than resolution





Resolution | Reciprocal- and real-space

- Reciprocal space: Peak width, *dQ*
  - PDF collections often maximise Q_{max} at expense of *dQ*



- Real space: Damping with *r*



Resolution | Reciprocal- and real-space

- Reciprocal space: Peak width, *dQ*



  - PDF collections often maximise Q_{max} at expense of *dQ*
  - This can make everything look like a nanoparticle!
  - Normal rules apply: Measure a standard!

- Real space: Low *r* region is unaffected by *dQ*



Resolution | dQ_damping

– Damping due to a constant *dQ* is easily modelled

```
macro dQ_damping(dQ,dQv)
{
   'Models dampening of the PDF intensity as a function of r due to a constant width reciprocal space peak shape
   'Ref: Acta Cryst. (1992). A48, 336-346
   #m_argu dQ 'Instrumental FWHM of S(Q) data
   If_Prm_Eqn_Rpt(dQ, dQv, min 0.001 max 0.2, del 0.0001)
   scale_phase_X = Exp(-0.5  X^2 (CeV(dQ,dQv)/2.35482)^2);
}
```

– dQ is a parameter defining the Gaussian FWHM instrumental width of the Bragg data, in *Q*

– dQ should be refined using a standard with no sample broadening (e.g. Si) and then fixed

– This should be applied at the **xdd** level

– Usage:

```
xdd filename.xy
   dQ_damping(@,0.05)
```



Resolution | dQ_lor_damping

- Bragg peak shapes are rarely pure Gaussians
- A modification models a Lorentzian component to a Pseudo-Voight Bragg peak shape

```
macro dQ_lor_damping(dQ,dQv,lor,lorv)
{
   'A modification of dQ_dampening to correct for a Lorentzian component to the reciprocal space peak shape
   #m_argu dQ 'Instrumental FWHM of S(Q) data
   #m_argu lor 'Lorentzian contribute to peak shape (between 0 and 1)
   If_Prm_Eqn_Rpt(dQ, dQv, min 0.001 max 0.2, del 0.0001)
   If_Prm_Eqn_Rpt(lor, lorv, min 0 max 1)
   scale_phase_X = (1-CeV(lor,lorv)) Exp(-0.5  X^2 (CeV(dQ,dQv)/2.35482)^2) + CeV(lor,lorv) Exp(-0.5 CeV(dQ,dQv) X);
}
```

  - `lor` is the Lorentzian fraction in the Bragg data
  - `dQ` and `lor` should be refined using a standard with no sample broadening (e.g. Si) and then fixed
  - Usage:

```
xdd filename.xy
     dQ_lor_damping(dQ,0.05,lor,0.1)
```

Resolution | `convolute_alpha`

- If the Bragg FWHM increases as a function of *Q*, this causes *r*-dependent broadening[†]

$$\mathrm{FWHM}(Q) = c + \alpha Q$$

where *c* is our constant FWHM, `dQ`

- This is modelled with a convolution

```
macro convolute_alpha(alpha,alphav)
{
   'Convolution to account for a Q dependent broadening term in the S(Q) data
   #m_argu alpha
   If_Prm_Eqn_Rpt(alpha, alphav, min 0.00001 max 0.1, del 0.001)
   local #m_unique fwhm = CeV(alpha,alphav) Xo;
   pdf_convolute = Gauss(0, fwhm);
      min_X = Min( Ln(0.001) / fwhm,0) ;
      max_X = Max(-Ln(0.001) / fwhm,0) ;
      convolute_X_recal = If(Xo,1,1) 1;
}
```

- Usage:
```
xdd filename.xy
    convolute_alpha(!alpha,0.001)
```

[†] X. Qiu, E. S. Bozin, P. Juhas, T. Proffen and S. J. L. Billinge, *J. Appl. Cryst.* (2004) **37**, 110-116

Resolution | pkshape_dQ_alpha

- A Bragg peak shape using dQ, lor and alpha can be used to extract instrumental parameters from Bragg data of a standard

```
macro pkshape_dQ_alpha(dQ,dQv,alpha,alphav,lor,lorv)
{
   'Bragg peak shape of the form (dQ^2 + alpha^2 Q^2)^0.5 for 2theta data
   '  Q = (2 Pi/D_spacing)
   '  deltaQ/Q = (dQ^2 + alpha^2 Q^2)^0.5 / Q
   '  delta2Th = 2 deltaQ/Q Tan(Th)
   #m_argu dQ     'Fixed peak width, in Q
   #m_argu alpha  'Q dependent peak width term
   #m_argu lor    'Lorch fraction
   If_Prm_Eqn_Rpt(dQ, dQv, min 0.001 max 0.1, del 0.0001)
   If_Prm_Eqn_Rpt(alpha, alphav, min 0.00001 max 0.01, del 0.001)
   If_Prm_Eqn_Rpt(lor, lorv, min 0 max 1)
   peak_type pv
      pv_lor = CeV(lor,lorv);
      pv_fwhm = 2 (CeV(dQ,dQv)^2 + CeV(alpha,alphav)^2 (2 Pi/D_spacing)^2)^0.5 / (2 Pi/D_spacing)  Tan(Th) Rad;
}
```

- This does a decent job of fitting Bragg data
- These can then be fixed and used in PDF refinements

Resolution | pkshape_dQ_alpha

- A joint refinement can be performed using the same instrumental parameters in PDF and Bragg datasets when using this peak shape
  - Outline .inp file:

```
prm dQ  0.05 prm alpha  0.002 prm lor  0.1 prm Qmax 30

xdd pdf_filename.xy
   dQ_lor_damping( ,dQ, ,lor)
   convolute_Qmax_Sinc( ,Qmax)
   convolute_alpha( ,alpha)

   str phase_name "Si"
      Cubic(aLP 5.4311946)
      space_group "Fd-3m"
      site Si  x 0.  y 0.  z 0.  occ Si 1.  beq_rcut_rlo_spherical(!rcut,3,@,0.1,!rlo,4.2,@,0.4, ,beqhi,@,5)

xdd Bragg_filename.xy
   LP_Factor( ,90.0)
   lam ymin_on_ymax 0.0001 la 1.0 lo =12.3984191/76.69; :  0.1616693 lh 0.1
   bkg @  0 0 0 0 0 0

   str phase_name "Si"
      Cubic(aLP 5.4311946)
      space_group "Fd-3m"
      site Si  x 0.  y 0.  z 0.  occ Si 1.  beq  beqhi  0.4
      pkshape_dQ_alpha( ,dQ, ,alpha, ,lor)
      scale @  4e-005
```

Resolution | `pkshape_dQ_alpha`

- A joint refinement can be performed using the same instrumental parameters in PDF and Bragg datasets when using this peak shape





PDF Processing Functions

PDF Processing | Q_{max} and PDF resolution

- Δr resolution of a PDF is dominated by Q_{max}
  - $Q = 2\pi/d = 4\pi\sin\theta/\lambda$
  - $\Delta r \approx 2\pi/Q_{max}$
  - Resolution is Q_{max} limited unless $Q_{max} > 3/\langle\langle u^2\rangle\rangle^{½}$ †

$AlPO_4$





†B. H. Toby and T. Egami, *Acta Cryst. A*, **48** (1992) 336

PDF Processing | Q_{max} Sinc function

- The PDF is produced from the *F*(*Q*) data by the Fourier transform (FT)†

$$G(r) = \frac{1}{(2\pi)^3 \rho_0} \int_0^\infty 4\pi Q^2 F(Q) \frac{\sin Qr}{Qr} \, \mathrm{d}Q$$

- Ideally the FT is done from zero to infinity
- In practice data is only collected over a finite *Q* range, from Q_{min} to Q_{max}

†D. Keen, *J. Appl. Cryst.* (2001) **34**, 172

PDF Processing | Q_{max} Sinc function

- Fourier transform of the $F(Q)$ over a finite Q range gives rise to termination ripples in the PDF
- FT of a box function is a sinc function





J. S. Chung and M. F. Thorpe, *Phys. Rev. B* (1997) **55**(3), 1545.

PDF Processing | convolute_Qmax_Sinc

- The sinc convolution can be modelled

```
macro convolute_Qmax_Sinc(Qmax,Qmaxv)
{
   'Convolution to include Sinc function due to termeination of S(Q) at finite Qmax during Fourier transform
   'Use in combination with start_X 1 to avoid very low-r data
   'Ref: J. S. Chung and M. F. Thorpe, Phys. Rev. B, 55(3), 1997, 1545
   #m_argu Qmax
   If_Prm_Eqn_Rpt(Qmax, Qmaxv, min 1 max 100, del 0.01)
   local #m_unique conv_max = (5 CeV(Qmax,Qmaxv) - Mod(5 CeV(Qmax,Qmaxv),1))/5 2 Pi / CeV(Qmax,Qmaxv);
   pdf_convolute = If(Abs(X) > Yobs_dx_at(Xo),(Sin(CeV(Qmax,Qmaxv) X)/(X)),CeV(Qmax,Qmaxv));
      min_X = Min(-Xo,-conv_max) ;
      max_X = Max( Xo, conv_max) ;
}
```

- `Qmax` is the only parameter
  - `Qmax` should be fixed at the known Q_{max} value used, but can be refined if needed
- This should be applied at the **xdd** level
  - Usage:

```
xdd filename.xy
   convolute_Qmax_Sinc(!Qmax,30)
```

PDF Processing | `convolute_Qmax_Sinc`

- `convolute_Qmax_Sinc` replicates the ripples in a PDF generated to a finite Q_{max}



- It currently fails at low-r, so use `start_X 1`

PDF Processing | Modification functions

- Termination ripples can be mitigated against with a Lorch (or Soper-Lorch[†]) function
  - Applied at the Fourier transform step



$$L_0(Q, \Delta_0) = (\sin Q\Delta_0)/(Q\Delta_0) = j_0(Q\Delta_0)$$

- Done at the expense of real-space resolution

$$\mathrm{FWHM} = [d_0/\ln(2)]\ln(2\cosh(\,(d_1/d_0)\ln(2)r\,)\,)$$

[†]A. K. Soper and E. R. Barney, *J. Appl. Cryst.* (2012) **45**, 1314–1317

PDF Processing | Modification functions

– Termination ripples can be mitigated against with a Lorch (or Soper-Lorch†) function



– Done at the expense of real-space resolution

$$\mathrm{FWHM} = [d_0/\ln(2)]\ln(2\cosh(\,(d_1/d_0)\ln(2)r\,)\,)$$

†A. K. Soper and E. R. Barney, *J. Appl. Cryst.* (2012) **45**, 1314–1317

PDF Processing | `convolute_Lorch`

- Lorch broadening is modelled with

```
macro convolute_Lorch(Qmax,Qmaxv)
{
   'Convolution to account for the use of a Lorch function prior to Fourier transform of the S(Q)
   #m_argu Qmax
   If_Prm_Eqn_Rpt(Qmax, Qmaxv, min 5 max 60, del 0.1)
   local #m_unique fwhm = (Pi / CeV(Qmax,Qmaxv))/Ln(2);
   pdf_convolute = Gauss(0, fwhm);
      min_X = Ln(0.001) / fwhm;
      max_X =-Ln(0.001) / fwhm;
      convolute_X_recal = If(Xo,1,1) 1;
}
```

- For the original Lorch function `Qmax` is the only parameter
  - `Qmax` should be fixed at the known Q_{max} value as it is correlated with beq values
- This should be applied at the **`xdd`** level
  - Usage:

```
xdd filename.xy
   convolute_Lorch(!Qmax,30)
```

PDF Processing | convolute_SoperLorch

- Soper-Lorch broadening is modelled with

```
macro convolute_SoperLorch(d_zero,d_zerov,d_one,d_onev)
{
   'Covolution to account for the use of the Soper-Lorch function prior to Fourier transform of the S(Q)
   'd_zero is the "Width of broadening in r space", as used in GudrunN
   'Ref: A. K. Soper and E. R. Barney, J. Appl. Cryst. (2012). 45, 1314-1317
   #m_argu d_zero
   #m_argu d_one
   If_Prm_Eqn_Rpt(d_zero, d_zerov, min 0.001 max 1.0, del 0.01)
   If_Prm_Eqn_Rpt(d_one, d_onev, min 0 max 0.1, del 0.01)
   local #m_unique fwhm = 2.3548 (CeV(d_zero,d_zerov)/Ln(2) Ln(2 Cosh( (CeV(d_one,d_onev)/CeV(d_zero,d_zerov)) Ln(2) Xo)));
   pdf_convolute = Gauss(0, fwhm);
      min_X = Ln(0.001) / fwhm;
      max_X =-Ln(0.001) / fwhm;
      convolute_X_recal = If(Xo,1,1) 1;
}
```

```
macro convolute_SoperLorch(d_zero,d_zerov)
{
   convolute_SoperLorch(d_zero,d_zerov,!d_one,0)
}
```

- d_zero is the 'Width of broadening in r space' when this function is used within Gudrun
  - d_zero should be fixed in the refinement as it is correlated with beq values
  - d_one is available, but is not often fixed at zero
- This should be applied at the **xdd** level
  - Usage:

```
xdd filename.xy
   convolute_SoperLorch(!d_zero,0.1)
```

PDF Peak Shape Functions

PDF Peak Shape | Theory

- The Gaussian FWHM of a PDF peak between atoms A and B is $\surd(B_{iso,A}\ B_{iso,B}\ \mathrm{Ln}(2) / \pi^2)$
  - B_{iso} is a measure of thermal motion of each atom
- In TOPAS, the B_{iso} (`beq`) values are used to determine the observed peak width
  - These can be a defined as a function of *r*

- NOTE: Strictly speaking the term B_{iso} should only be used at high *r*, where motion is completely uncorrelated

PDF Peak Shape | beq_rcut

- PDF peak are narrower at low-*r* due to correlated motion
- A simple peak shape can differentiate between values below and above a given cutoff *r* value

```
macro beq_rcut(rcut,beqlo,beqlov,beqhi,beqhiv)
{
   #m_argu beqlo
   #m_argu beqhi
   #m_argu rcut
   If_Prm_Eqn_Rpt(beqlo, beqlov, min 0.001 max 10, del = 0.05 Val;)
   If_Prm_Eqn_Rpt(beqhi, beqhiv, min 0.001 max 10, del = 0.05 Val;)
   beq = IF X < rcut THEN
            CeV(beqlo, beqlov)
         ELSE
            CeV(beqhi, beqhiv)
         ENDIF;
}
```



- This replaces the **beq** part of a **site**

```
site Si   x 0    y 0    z 0    occ Si 1.    beq_rcut(3,beqlo,0.1,beqhi,0.5)
```

PDF Peak Shape | `beq_spherical`

- The transition from correlated to uncorrelated motion is not normally a step function
- This macro uses the PDF of a sphere to scale between a **`beq`** at low-*r* (correlated) to one at high-*r* (uncorrelated motion)
  - `beqlo`, `beqhi` and the `radius` of the sphere are refined

```
site Si  x 0  y 0  z 0  occ Si 1.  beq_spherical(beqlo,0.1,beqhi,0.5,radius,10)
```



PDF Peak Shape | beq_rcut_rlo_spherical

- This macro uses multiple cutoffs and a sphere radius for a transition from correlated (low-*r*) to uncorrelated (high-*r*) motion

```
macro beq_rcut_rlo_spherical(rcut,rcutv,beqcut,beqcutv,rlo,rlov,beqlo,beqlov,beqhi,beqhiv,radius,radiusv)
{
   #m_argu rcut
   #m_argu beqcut
   #m_argu rlo
   #m_argu beqlo
   #m_argu beqhi
   #m_argu radius
   If_Prm_Eqn_Rpt(rcut, rcutv, min 0.1 max 10, del = 0.05 Val;)
   If_Prm_Eqn_Rpt(beqcut, beqcutv, min 0.001 max 10, del = 0.05 Val;)
   If_Prm_Eqn_Rpt(rlo, rlov, min 0.1 max 100, del = 0.05 Val;)
   If_Prm_Eqn_Rpt(beqlo, beqlov, min 0.001 max 10, del = 0.05 Val;)
   If_Prm_Eqn_Rpt(beqhi, beqhiv, min 0.001 max 100, del = 0.05 Val;)
   If_Prm_Eqn_Rpt(radius, radiusv, min 1 max 1000, del = 0.05 Val;)
   beq =
      IF X > (2 CeV(radius, radiusv)) + CeV(rlo, rlov) THEN
         CeV(beqhi, beqhiv)
      ELSE
         IF X > CeV(rlo, rlov) THEN
            CeV(beqlo, beqlov) + (CeV(beqhi, beqhiv)-CeV(beqlo, beqlov))
               (1-(Pi (X-CeV(rlo, rlov))^2 ((0.25 ((X-CeV(rlo, rlov))/CeV(radius, radiusv))^3)-(3 (X-CeV(rlo, rlov))/CeV(radius, radiusv))+4)/(4 Pi (X-CeV(rlo, rlov))^2)))
         ELSE
            IF X > CeV(rcut, rcutv) THEN
               CeV(beqlo, beqlov)
            ELSE
               CeV(beqcut, beqcutv)
            ENDIF
         ENDIF
      ENDIF;
}
```



- Usage:

```
site Si  x 0  y 0  z 0  occ Si 1.
   beq_rcut_rlo_spherical(!rcut,2,beqcut,0.1,!rlo,5,beqlo,0.2,beqhi,0.5,radius,10)
```

PDF Peak Shape | `beq_rcut_rlo_spherical`

- This gives great flexibility, but take care to reduce the number of refined parameters
  - Constrain `beqcut` to be the same for all/most atoms
  - Constrain `radius` to be the same for all atoms
  - Are `rlo` *and* `rcut` needed?



  - Constrain `beq` values for atoms of the same type to the same value

```
site Si  x 0  y 0  z 0  occ Si 1.
| beq_rcut_rlo_spherical(!rcut,2,beqcut,0.1,!rlo,5,beqlo,0.2,beqhi,0.5,radius,10)
```

PDF Peak Shape | beq_PDFfit2

– The peak shape used in PDFfit2 and PDFgui has been reproduced in a TOPAS macro

```
macro beq_PDFfit2(uiso,uisov,rcut,rcutv,sratio,sratiov,delta1,delta1v,delta2,delta2v,qbroad,qbroadv)
{
   #m_argu uiso
   #m_argu rcut
   #m_argu sratio
   #m_argu delta1
   #m_argu delta2
   #m_argu qbroad
   If_Prm_Eqn_Rpt(uiso, uisov,  min 0.000001 max 1, del = 0.01 Val;)
   If_Prm_Eqn_Rpt(rcut,  rcutv,   min 0.000001 max 10, del = 0.01 Val;)
   If_Prm_Eqn_Rpt(sratio,sratiov, min 0.000001 max 10, del = 0.01 Val;)
   If_Prm_Eqn_Rpt(delta1,delta1v, min 0.000001 max 1, del = 0.01 Val;)
   If_Prm_Eqn_Rpt(delta2,delta2v, min 0.000001 max 1, del = 0.01 Val;)
   If_Prm_Eqn_Rpt(qbroad,qbroadv, min 0.000001 max 1, del = 0.01 Val;)
   beq = If(X < CeV(rcut, rcutv), 0.5 CeV(sratio, sratiov), 1) CeV(uiso, uisov)
   8 Pi^2 (1 - Min(((CeV(delta1, delta1v)/X) + (CeV(delta2, delta2v)/X^2) - (CeV(qbroad, qbroadv)^2 X^2)),1));
}
```

– Refinable parameters are named as they appear in PDFgui

PDF Peak Shape | `beq_PDFfit2`

- Usage:

```
site Si  x 0  y 0  z 0  occ Si 1.
   beq_PDFfit2(@,0.001,!rcut,2,sratio,0.5,delta1,0.001,delta2,0.001,!qbroad,0.05)
```

- Note that a U_{iso} is used, rather than B_{iso}
  - $B_{iso} = 8\pi^2 U_{iso}$
- Sharpening parameters `sratio`, `delta1` and `delta2` should be fixed between atoms in the same phase
- `qbroad` should be determined using a standard, and then fixed
- Currently TOPAS does not handle anisotropic thermal parameters

Summary | Features of a PDF revisited

- Peak positions: Bond lengths, geometry, unit cell
- Peak widths: Vibrations, phonons, disorder
- Peak areas: Coordination number, occupancies
- Peak dampening: Size, shape



All of these can be modelled using TOPAS v6

Summary | Useful relationships

- Bragg ⟷ PDF





- Constant peak width ⟷ Damping with *r*
- *αQ* peak broadening ⟷ Peak broadening with *r*
- Displacement parameters ⟷ Peak width
- Particle size broadening ⟷ Damping with *r*
- Fourier transform ⟷ Sinc function oscillations
- FT with Lorch function ⟷ Constant peak broadening

- For TOF data, see I.-K. Jeong, M. J. Graf and R. H. Heffner *J. Appl. Cryst.* (2005). **38**, 55–61